
Reproducible and automated reporting

Stata’s commands for report generation allow you to create

complete documents that include formatted text, summary

statistics, regression results, and graphs in any of the following

formats:

Word R©

Excel R©

PDF

HTML

K. L. MacDonald (StataCorp) 30 August 2019 2 / 74



Reproducible and automated reporting

Stata’s commands for creating reports come in two varieties:

1 Dynamic document commands–dyntext and dyndoc

These commands create text files, HTML files, and Word

documents that incorporate the full output from Stata

commands. You can use the Markdown text-formatting

language to customize the look of your report.

2 put* commands–putdocx, putpdf, and putexcel

These commands create Word documents, PDFs, and Excel

files that insert results from Stata commands into formatted

text and tables in your document.

Whether you choose the dynamic document commands or the

put* commands, you can create documents that are reproducible

and dynamic.

Reproducible

Reproducibility is at the core of Stata’s design.

Use the version 16 command, and any commands you run

today will produce the same results many years from now.

Use the datasignature command, you can verify that your

data have not changed.

Incorporate these tools for reproducibility into the do-files or

text files that create your reports, and those reports will also

be reproducible. Rerun your commands at any time and

re-create your report.

Dynamic

Update reports as data changes.

Say you need to run the same report monthly. Rerun the

commands that created the report with the updated dataset.

All Stata results in the report are updated automatically.

A whirlwind tour:

Word documents with putdocx

PDFs with putpdf

Excel files with putexcel

HTML documents with dyndoc

Word documents with dyndoc

Converting files from one type to another

Content for our report:

We have fictional data on the long-term care choices of 151

patients.

Each patient selected a nursing home, in-home care, or an

assisted-living facility.

We fit a conditional logistic regression model to model the

choices using cmclogit.

We include two case-specific covariates: marital status and a

five-category health status rating.

We also include two alternative-specific covariates: monthly

cost and distance from nearest relative.

. use ltcare, clear

. cmset id carealt

caseid variable: id
alternatives variable: carealt

. cmtab, choice(chosen)

Tabulation of chosen alternatives (chosen = 1)

Long-term care
alternatives Freq. Percent Cum.

Nursing home 58 38.41 38.41
In-home care 63 41.72 80.13

Assisted living 30 19.87 100.00

Total 151 100.00

. cmclogit chosen cost reldist, casevars(i.married i.hlthstat)

Iteration 0: log likelihood = -119.36634
Iteration 1: log likelihood = -87.257826
Iteration 2: log likelihood = -80.030792
Iteration 3: log likelihood = -79.791309
Iteration 4: log likelihood = -79.790189
Iteration 5: log likelihood = -79.790189

Conditional logit choice model Number of obs = 453
Case ID variable: id Number of cases = 151

Alternatives variable: carealt Alts per case: min = 3
avg = 3.0
max = 3

Wald chi2(12) = 52.13
Log likelihood = -79.790189 Prob > chi2 = 0.0000

chosen Coef. Std. Err. z P>|z| [95% Conf. Interval]

carealt
cost -.8927349 .1747136 -5.11 0.000 -1.235167 -.5503025

reldist -.005768 .0036172 -1.59 0.111 -.0128575 .0013215

Nursing_home (base alternative)

In_home_care
married

Married 3.682328 .7946326 4.63 0.000 2.124877 5.23978

hlthstat
Poor .7347672 1.346105 0.55 0.585 -1.90355 3.373084
Fair 4.894007 1.462341 3.35 0.001 2.027871 7.760143

Average 6.52018 1.751774 3.72 0.000 3.086766 9.953594
Good 4.800188 1.780498 2.70 0.007 1.310477 8.289899

_cons -3.44643 1.313259 -2.62 0.009 -6.020372 -.8724894

Assisted_living
married

Married 3.822477 .876701 4.36 0.000 2.104175 5.54078

hlthstat
Poor -1.573831 1.603339 -0.98 0.326 -4.716317 1.568655
Fair 4.180285 1.423912 2.94 0.003 1.389469 6.971101

Average 6.179002 1.707872 3.62 0.000 2.831634 9.526369
Good 4.689159 1.72597 2.72 0.007 1.306321 8.071998

_cons -4.314829 1.298062 -3.32 0.001 -6.858984 -1.770674

We want to evaluate what would happen if cost of care in

nursing homes increases by 15%.

We use margins to estimate changes in the expected

proportions of individuals choosing each long-term care option.

. margins, alternative(1) at(cost=generate(cost)) at(cost=generate(cost*1.15))

Predictive margins Number of obs = 453
Model VCE : OIM

Expression : Pr(carealt|1 selected), predict()
Alternative : Nursing home

1._at : cost = cost

2._at : cost = cost*1.15

Delta-method
Margin Std. Err. z P>|z| [95% Conf. Interval]

_outcome#_at
Nursing home#1 .384106 .021637 17.75 0.000 .3416982 .4265137
Nursing home#2 .3320989 .0254219 13.06 0.000 .2822728 .381925
In-home care#1 .4172185 .030231 13.80 0.000 .3579668 .4764702
In-home care#2 .4518933 .0332705 13.58 0.000 .3866843 .5171024

Assisted living#1 .1986755 .0265481 7.48 0.000 .1466422 .2507088
Assisted living#2 .2160078 .0291166 7.42 0.000 .1589404 .2730752

. marginsplot, xdim(_outcome) ///
> legend(order(1 "Current cost" 2 "15% increase") cols(1) ring(0) ///
> bplacement(neast)) ylabel(.1(.1).6) ///
> title("Effect of 15% increase in nursing-home cost") ///
> ytitle(Expected probability) xtitle(" ") noci graphregion(margin(r=10))

Variables that uniquely identify margins: _atopt _outcome
Multiple at() options specified:

_atoption=1: cost=generate(cost)
_atoption=2: cost=generate(cost*1.15)

.

. graph export cost.png, replace
(file cost.png written in PNG format)

. margins, alternative(1) at(cost=generate(cost)) at(cost=generate(cost*1.15)) ///
> contrast(at(r) nowald)

Contrasts of predictive margins Number of obs = 453
Model VCE : OIM

Expression : Pr(carealt|1 selected), predict()
Alternative : Nursing home

1._at : cost = cost

2._at : cost = cost*1.15

Delta-method
Contrast Std. Err. [95% Conf. Interval]

_at@_outcome
(2 vs 1) Nursing home -.0520071 .0107312 -.0730398 -.0309743
(2 vs 1) In-home care .0346748 .0075127 .0199501 .0493994

(2 vs 1) Assisted living .0173323 .0067909 .0040223 .0306423

putdocx

Basic putdocx structure

putdocx begin

putdocx save myfile.docx

Basic putdocx structure

putdocx begin

putdocx paragraph
putdocx text = ("mytext")

putdocx save myfile.docx

Basic putdocx structure

putdocx begin

putdocx paragraph
putdocx text = ("mytext")

putdocx textblock begin
mytext

putdocx textblock end

putdocx save myfile.docx

Basic putdocx structure

putdocx begin

putdocx paragraph
putdocx text = ("mytext")

putdocx textblock begin
mytext

putdocx textblock end

putdocx paragraph
putdocx image myimage.png

putdocx save myfile.docx

Basic putdocx structure

putdocx begin

putdocx paragraph
putdocx text = ("mytext")

putdocx textblock begin
mytext

putdocx textblock end

putdocx paragraph
putdocx image myimage.png

putdocx table tablename = etable

putdocx save myfile.docx

We can start with this basic structure and add to it to create

a report with the results from our choice model.

We begin our document and add a header with the Stata

logo. We also put page numbers in the footer.

version 16
putdocx begin, header(head1) footer(foot1)

putdocx paragraph, toheader(head1) halign(right)
putdocx image stata16logo.png, height(.2in)

putdocx paragraph, tofooter(foot1)
putdocx pagenumber

We add a heading and an introduction to our report.

putdocx paragraph, style(Heading1)
putdocx text ("Introduction")

putdocx textblock begin
We are interested in studying the determinants of long-term care
choices. We are also concerned that the cost of nursing home care
is likely to increase by approximately 15%, and we want to know the
expected impact of this change on the probability of individuals
selecting each type of long-term care available.
putdocx textblock end

We add a description of our data.

use ltcare, clear
cmset id carealt
local ncase = r(n_cases)
total chosen, over(carealt)

putdocx textblock begin
In our sample of <<dd_docx_display: ‘ncase’>> patients,
<<dd_docx_display: _b[c.chosen@1bn.carealt]>>
selected a nursing home,
<<dd_docx_display: _b[c.chosen@2.carealt]>>
selected in-home care, and
<dd_docx_display: _b[c.chosen@3.carealt]>> selected an
assisted-living facility.
putdocx textblock end

We describe our model and add a table of the results.

putdocx paragraph, style(Heading1)
putdocx text ("Model")

putdocx textblock begin
We fit a conditional logistic regression model with
alternative-specific covariates monthly cost and distance from
nearest care-giving relative. We also include case-specific
covariates marital status and health status.

The results are presented in Table 1:
putdocx textblock end

cmclogit chosen cost reldist, casevars(i.married i.hlthstat)
cmclogit, cformat(%5.3f) pformat(%4.2f) sformat(%5.3f)
putdocx table cmcl = etable, title("Table 1")

We present the results of our margins command graphically.

putdocx paragraph, style(Heading1)
putdocx text ("Effect of increasing cost of care in nursing homes")

margins, alternative(1) at(cost=generate(cost)) ///
at(cost=generate(cost*1.15))

marginsplot, xdim(_outcome) ///
legend(order(1 "Current cost" 2 "15% increase") ///
cols(1) ring(0) bplacement(neast)) ylabel(.1(.1).6) ///
title("Effect of 15% increase in nursing-home cost") ///
ytitle(Expected probability) xtitle(" ") noci ///
graphregion(margin(r=10)))

graph export cost.png, replace

putdocx paragraph, halign(center)
putdocx image cost.png, height(3in)

We summarize the expected changes in probabilities when for

increased nursing-home prices.

Finally, we save our document.

margins, alternative(1) at(cost=generate(cost)) ///
at(cost=generate(cost*1.15)) ///
contrast(at(r) nowald)

putdocx textblock begin
If nursing home costs increase by 15%, we anticipate a
<<dd_docx_display: -‘:display %3.1f 100*r(b)[1,1]’>>
percentage point decrease in the number of individuals selecting
long-term care in nursing homes. In turn, we expect the number of
individuals selecting in-home care to increase by
<<dd_docx_display: ‘:display %3.1f 100*r(b)[1,2]’>> and the number
of individuals selecting assisted-living facilities to increase by
<<dd_docx_display: ‘:display %3.1f 100*r(b)[1,3]’>>
percentage points.
putdocx textblock end

putdocx save myrpt, replace

putpdf

Basic putpdf structure

putpdf begin

putpdf save myfile.pdf

Basic putpdf structure

putpdf begin

putpdf paragraph
putpdf text = ("mytext")

putpdf save myfile.pdf

Basic putpdf structure

putpdf begin

putpdf paragraph
putpdf text = ("mytext")

putpdf paragraph
putpdf image myimage.png

putpdf save myfile.pdf

Basic putpdf structure

putpdf begin

putpdf paragraph
putpdf text = ("mytext")

putpdf paragraph
putpdf image myimage.png

putpdf table tablename = etable

putpdf save myfile.pdf

putexcel

Basic putexcel structure

putexcel set myfile.xlsx

Basic putexcel structure

putexcel set myfile.xlsx

putexcel A1 = "mytext"

Basic putexcel structure

putexcel set myfile.xlsx

putexcel A1 = "mytext"

putexcel A2 = image(myimage.png)

Basic putexcel structure

putexcel set myfile.xlsx

putexcel A1 = "mytext"

putexcel A2 = image(myimage.png)

putexcel B10 = etable

K. L. MacDonald (StataCorp) 30 August 2019 44 / 74



Reproducible and automated reporting

K. L. MacDonald (StataCorp) 30 August 2019 45 / 74



Reproducible and automated reporting

K. L. MacDonald (StataCorp) 30 August 2019 46 / 74



Reproducible and automated reporting

K. L. MacDonald (StataCorp) 30 August 2019 47 / 74



Reproducible and automated reporting

dyndoc

Basic dyndoc structure

Markdown-formatted text

Title
=====

Basic dyndoc structure

Markdown-formatted text

Title
=====

##Subtitle

Basic dyndoc structure

Markdown-formatted text

Title
=====

##Subtitle

text with **bolding**

K. L. MacDonald (StataCorp) 30 August 2019 51 / 74



Reproducible and automated reporting

Basic dyndoc structure

Markdown-formatted text

Title
=====

##Subtitle

text with **bolding**

~~~
codeblock
~~~

Basic dyndoc structure

Dynamic tags

<<dd_do>>
Stata commands

<</dd_do>>

Basic dyndoc structure

Dynamic tags

<<dd_do>>
Stata commands

<</dd_do>>

<<dd_display: expression>>

Basic dyndoc structure

Dynamic tags

<<dd_do>>
Stata commands

<</dd_do>>

<<dd_display: expression>>

<<dd_graph:saving(mygraph.png) graphname(mygraph)>>

Convert the myhtml.md Markdown document to HTML.

. dyndoc myhtml.md

Convert the myhtml.md Markdown document to a Word

document.

. dyndoc myhtml.md, docx

Or convert the HTML document directly to a Word document.

. html2docx myhtml.html

Incoroporate HTML code and style sheets to change the look of

the document.

We save the Markdown file that includes the header file as the

myhtml2.md. Then we can type

. dyndoc myhtml2.md

to convert it to an HTML file.

The same style is applied to a the Word document that is created

when we type

. dyndoc myhtml2.md, docx

A few edits make it easy to produce dynamic reports. For instance,

change

<<dd_do:quietly>>
use ltcare, clear
cmset id carealt
<</dd_do>>

to

<<dd_do:quietly>>
use ‘1’, clear
cmset id carealt
<</dd_do>>

Save the file with this update as mythmld.md, then type

. dyndoc mythmld.md ltcare_q3.dta

to create the same report using the ltcare q3.dta

Add more flexibility by passing additional arguments. For instance,

we can specify the percentage increase in nursing-home cost that

we want to analyze as a second argument. We change each

instance of

15%

within text to in our Markdown document to

<<dd_display:‘2’>>

We also change the margins and marginsplot to respect the

percentage specified in the second argument.

<<dd_do: quietly>>
margins, alternative(1) at(cost=generate(cost)) ///
at(cost=generate(cost*1.‘2’))

marginsplot, xdim(_outcome) ///
legend(order(1 "Current cost" 2 "‘2’% increase") ///
cols(1) ring(0) bplacement(neast)) ylabel(.1(.1).6) ///
title("Effect of ‘2’% increase in nursing-home cost") ///
ytitle(Expected probability) xtitle(" ") noci ///
graphregion(margin(r=10))

<</dd_do>>

After we save the modified file as myhtmld.md, we can type

. dyndoc myhtmld.md ltcare_q3.dta 20, saving(q3ptc20.html) replace

to evaluate the effect of a 20% increase in the cost nursing-home care and
create the HTML file q3pct20.html that reports the results.

Whirlwind tour of reporting features?

Tornado?

Cyclone?

Try putdocx, putpdf, putexcel, or dyndoc the next time you

need to automate a report in Word, HTML, Excel, or PDF

format.

